
ABSTRACT 
One of the most expensive aspects of collecting data is ascertaining the quality of the data collected. 
The focus of the California Perinatal Quality Care Collaborative (CPQCC) is the development of perinatal and 
neonatal outcomes and information, which allows for data driven performance improvement and benchmarking 
throughout California. 
The CPQCC base management and reporting system was designed to offer an efficient way to allow a) on-line data 
submission that would be convenient for CPQCC participants, b) instant error checking, and c) real-time reporting 
resulting in a high quality database supporting informed decisions. Also developed was a separate annual report 
module that allows the establishment and monitoring of QI goals. SAS®/BASE, SAS/IntrNet®, SAS/GRAPH®, MS 
Office/Access, MS Office/Excel, Winrar, Adobe Acrobat Distiller, and JavaScript were integrated to form this 
application. 
This presentation summarizes and demonstrates the on-line CPQCC database management and reporting system.  

INTRODUCTION 
The goal of the California Perinatal Quality Care Collaborative (CPQCC) is the development of perinatal outcomes 
and information in a data repository that would allow data driven performance improvement and benchmarking 
throughout California. Part of this goal was the creation of a CPQCC Master Database whose content was designed 
by experts from a variety of relevant backgrounds to provide a basis for setting and achieving QI goals. 
CPQCC was started prior to the more common usage of the WWW as a data collection tool. Two years ago when we 
were introduced to the project, the initiative had about 40 participating centers. Data were submitted either 
electronically or on paper forms. Data errors were a problem as well as data handling errors. For instance, the 
scanning process that was used to take in data submitted on paper forms, introduced errors on its own. The data 
correction process was elaborate and staff-intensive. A related problem was that errors could not be diagnosed in 
real-time. Rather there was a delay between receiving data and informing a hospital of the errors. The hospital had to 
go back and pull records and re-submit a new paper form or EDS record.  
Error reports, quarterly reports and annual summary reports were voluminous and took a lot of staff time to produce. 
Furthermore, the programming solutions were not always based on the same programming platform. 
As interest in the CPQCC initiative grew, it was anticipated that 80 or more additional centers would join resulting in a 
tripling of the work load of the data center. 

INFRASTRUCTURE 
CPQCC utilized several different applications: 

• Teleform was used to scan data submitted on paper; scanned data were transferred into an MS 
Office/Access database. 

• MS Office/Access was used to organize information on hospital contact information, the scanned paper 
forms, and some electronically submitted data. 

• Comma separated ASCII files were used to organize some electronically submitted data. 
• MS Office/Excel and MS Office/Word in conjunction with Java and/or SAS were used to compile error and 

quarterly reports. 
• Adobe Acrobat Writer and Distiller were used to convert MS Office/Excel documents into PDF documents. 
• SAS was used to generate several research reports and provide customized data analyses. 

SOLUTION 
A WWW-based application was built to handle data base management and reporting tasks. A first web site 
http://www.cpqccdata.org was created to perform data base management tasks, error and quarterly reporting. A 
second web site http://www.cpqccreport.org was created to handle the voluminous annual reports. The decision to 
use two different web addresses was based on the different audiences targeted: Data base management functions 
were targeted at hospital staff handling data entry while annual reporting functions were targeted at executive level 
staff seeking tools to develop and monitor QI initiatives. 
This solution required several additional applications: 

• MySQL and PHP were used to handle the password protected login mechanism of member hospitals to the 
data base management web site. 

• JavaScript was used for interacting with the user on the web site. It was one primary tool to accomplish real-
time error checking of data prior to its submission. 

• SAS/IntrNet was used to handle all tasks that involved access to the CPQCC Master DB. 
Due to problems with platform dependence, the role of Java was greatly reduced.  
The remaining sections demonstrate the data base management and reporting web sites further. Included in this 
paper are also code samples that show ways in which SAS and other applications were combined to allow smooth, 
relatively CPQCC staff independent data collection and reporting. 

WWW-BASED DATA BASE MANAGEMENT 

 
Figure 1: CPQCC DB Management Start Up Screen 

 
To further explain the components of the web-based data base management, Figure 1 shows the start up screen that 
is displayed to the hospital user upon successful login. Note that you can access a demonstration version of the 
CPQCC DB Management and Reporting web site by going to http://www.cpqccdata.org and supplying the username 
0000 and the password test. You can demo all the functionalities of the website that are open to the hospital user, any 
data submitted for the demonstration hospital 0000 are ignored for executive and network reports. 

ADD NEW DATA 
To add records for a new baby that was admitted to a member hospital certain eligibility criteria have to be met. These 
criteria are checked first. If a baby is eligible, the user has to assign a unique and new ID to this baby. If this baby died 
in the delivery room, a shorter form needs to be filled out. Once the user submits the information, the application 
server is contacted to verify whether the infant ID is already present in the CPQCC Master DB. If it is not, an empty 
form is loaded; otherwise, the form is loaded and the information already submitted retrieved from the CPQCC Master 
DB. 
The code below is used to determine whether the baby’s ID is already present in the CPQCC Master DB. It is a small 
part of the “gatekeeper” program that determines how a request from the web server is handled by the application 
server. The macro variable &ToDo has to be submitted with each request from the web server. It determines what 
output the application server produces. 

[ ... ]  
%ELSE %IF &ToDo EQ %STR(lookupID) %THEN %DO; /* check whether ID exists for 
                                                hospital number */ 
        %IF &byear EQ prevyear %THEN %LET 
            CohortYear=%EVAL(%SYSFUNC(YEAR(%SYSFUNC(today()))) - 1); 
        %ELSE %LET CohortYear=%SYSFUNC(YEAR(%SYSFUNC(today()))); 
        PROC SQL NOPRINT; 
             SELECT COUNT(*) 
                    INTO :nobsCHECK 
                    FROM cpqcc.CPQCCmaster (READ=xxxxxxx  
                    WHERE=(_recent EQ 'Y' AND id EQ &id AND hospno EQ &hospno)); 
        QUIT; 

        %IF &nobsCHECK > 0 %THEN %DO; 
            %loadIDWWW; 
        %END; 
        %ELSE %DO; 
            %addDataWWW; 
        %END; 
  %END; 
  [...] 
  
If the macro variable &nobsCHECK is 0, the application server runs the macro %addDataWWW which generates an 
empty form to enter data in. 
 
/*  
Macro loads an empty form into the browser for the hospital number, baby ID, and 
birth year that was provided by the user on the previous screen. 
Note that the largest part of the form is loaded from the file 
       DataEntryForm.html 
stored in the directory 
       c:\...\CPQCC\BETA\DB management\maclib 
The path to all pictures in this file is HARD coded to the root directory of the web 
server. Make sure that the pix directory is present in the root of the web server. 
*/ 
%MACRO addDataWWW; 
%GLOBAL SubmitType CohortYear id hospno commonwwwroot deldie keycode cpqccroot; 
DATA _NULL_; 
  FILE _webout; 
%IF &deldie EQ 0 %THEN %DO; 
  INFILE "&cpqccroot\DB management\maclib\DataEntryForm.html"  
          LRECL=257 PAD END=lastobs; 
%END; 
%ELSE %DO; 
  INFILE "&cpqccroot\DB management\maclib\DataEntryFormDRD.html"  
         LRECL=257 PAD END=lastobs; 
%END; 
  IF _N_ EQ 1 THEN DO; 
    PUT '<!DOCTYPE html PUBLIC "-//W3C//DTD XHTML 1.0 Transitional//EN"'; 
    PUT '"http://www.w3.org/TR/xhtml1/DTD/xhtml1-transitional.dtd">'; 
    PUT '<html xmlns="http://www.w3.org/1999/xhtml">'; 
    PUT '<head>'; 
    PUT '<title>CPQCC DB Management</title>'; 
    PUT '<meta name="Author" content="Beate Danielsen"></meta>'; 
    PUT '<meta http-equiv="Content-Type" content="text/html; charset=ISO-8859- 
         5"></meta>'; 
    PUT '<meta name="date" content="2000-04-05T10:52:00+00:00"></meta>'; 
    PUT '<link rel="stylesheet" href="' "&commonwwwroot.CSS/forms.css";  
    PUT '"       type="text/css"></link>'; 
    PUT '<style type="text/css">'; 
    [ ... more CSS style instructions deleted ... ] 
    PUT '</style>'; 
    PUT '<script language="JavaScript" type="text/javascript">'; 
    PUT '<!--'; 
    PUT 'function loadId(form) {'; 
    PUT '   form.hospno.value = "' "&hospno" '";'; 
    PUT '   form.hospno.readOnly = true;'; 
    PUT '   form.id.value     = "' "&id" '";'; 
    PUT '   form.id.readOnly = true;'; 
    PUT '   form.byear.value     = "' "&CohortYear" '";'; 
    PUT '   form.byear.readOnly = true;'; 
    PUT '   for (var i=1; i<parseInt(form.bdateyy.length); i++)'; 
    PUT '   {'; 
    PUT '     if ((form.bdateyy.options[i].text) == form.byear.value) '; 
    PUT ‘     {form.bdateyy.selectedIndex = i; i=parseInt(form.bdateyy.length);}'; 
    PUT '   }'; 
    PUT '   form.bdateyy.disabled = true;'; 

    PUT '   form.bdatemm.selectedIndex = 0;'; 
    PUT '   form.bdatedd.selectedIndex = 0;'; 
    PUT '}'; 
    PUT '// -->'; 
    PUT '</script>'; 
    [ ... more put statements to include JavaScript deleted ... ]     
    PUT '<script language="JavaScript" type="text/javascript">'; 
    PUT '<!--'; 
    PUT 'function allowSubmitErrors(form)'; 
    PUT '{'; 
    PUT '  if (form.allowErrors.checked == true) {'; 
    PUT '      form.allowErrorsHidden.value = "Y";'; 
    PUT '      form.ToDo.disabled = false;'; 
    PUT '      for(var i = 0; i < form.length; i++) {'; 
    PUT '      form.elements[i].disabled = false;'; 
    PUT '      }'; 
    PUT '  }'; 
    PUT '  else if (form.allowErrors.checked == false) formStatus(form,"N");'; 
    PUT '}'; 
    PUT '//-->'; 
    PUT '</script>'; 
    PUT '</head>'; 
    PUT '<body onload="initializeGlobals();' @@; 
    PUT '              DataCheck(document.DataEntry,' "'byear'" '); ' @@; 
    PUT '              loadId(document.DataEntry); ' @@; 
    PUT '              showMonth(document.DataEntry,' "'bdate'" ');">'; 
    PUT '<form method="post" name="DataEntry" action="' "&_URL" '">'; 
    PUT '<table>'; 
    PUT '<tr><th width="800" class="h1">CPQCC Data Management</th></tr>'; 
    PUT '<tr><th class="h2">New Infant ID - Add New Data On-Line</th></tr>'; 
  END; 
  PUT @1 thisline $CHAR256.; 
  IF lastobs THEN DO; 
     PUT '<input type="hidden" name="_service" value="' "&_SERVICE" '" />'; 
     PUT '<input type="hidden" name="keycode" value="' "&keycode" '" />'; 
     PUT '<input type="hidden" name="cpqccwwwroot" value="' "&cpqccroot" '" />'; 
     PUT '<input type="hidden" name="_debug" value="' "&_debug" '" />'; 
     PUT '</form>'; 
     PUT '</body>'; 
     PUT '</html>'; 
  END; 
RUN; 
%MEND addDataWWW; 
 

The %addDataWWW macro shows one of the important features we have implemented for the CPQCC Database 
Management web site. JavaScript is used to populate form data, and SAS is used to generate the appropriate 
JavaScript statements. The largest portion of the form is loaded from the appropriate HTML file based on the 
&deldie macro parameter (indicating whether or not the data pertain to a delivery room death) that is transmitted 
with the request from the website. Changes to the form can easily be implemented. The top part of the form is 
populated via the JavaScript function loadID and executed when the form is loaded into the browser as part of the 
<body onload="..."> tag. Finally, the form on the HTML page is properly initialized via the <form> tag. Variables 
needed for proper execution of the form are transmitted as hidden input elements. Figure 2 displays a part of the data 
entry form that is displayed. 

Figure 2: Data Entry Form loaded into the user’s browser 

 

 
Figure 3: Detail of Data Entry Screen 

Sections of the form open up upon clicking on the arrow next to the form section as is shown in Figure 3. Note that 
item 5 is already filled out as the user requested a form for a baby that did not die in the delivery room. 
JavaScript is used to verify that the form is filled out correctly and give immediate feedback to the user if an 
inconsistency is discovered. Only if the entire form is filled out correctly, it can be submitted. 

EDIT DATA 
Data can be edited by generating a list of IDs already present in the CPQCC Master DB for the member hospital 
currently logged on. This list is generated by the application server. It shows when the record was last updated, 
whether or not the record is complete, how many errors and how many unknown items were submitted for this ID.  

Figure 4: Edit Data Screen 
 

In Figure 4 above, a click on the ID number will load the appropriate data entry form to access the record and make 
corrections. Corrections can either be based on the dynamic error report (next section), or they can be based on the 
instant error check implemented using JavaScript. For instance, ID 20 in the list above shows 2 errors. Upon loading 
the record and performing the Data Check, a separate Windows pops up listing the two problems.  

 
Figure 5: JavaScript Based Error Check 

 
The two items are corrected. The record has to be checked for errors again. Only upon a successful check that no 
more errors are present, the data can be submitted. The result of the data submission is then displayed to the hospital 
user. For each item the variable, its description, its coded value, and its formatted value are shown. The top portion of 
the data submission result screen is displayed in Figure 6. Upon receiving the request for updating a record, the 
application server verifies the old record, flags it as old, then adds the new corrected record and flags it as the most 
recent record that should be used in all reporting functions. The record summary is printed by reading from the 
updated CPQCC Master DB, PUT statements are used to show the printed record detail. 

Figure 6: On-Line Data Submission Result (top part only) 

 

ERROR AND QUARTERLY REPORTS 
The error report does not only show errors, but also displays other data related to submissions the CPQCC Data 
Center has received from the member center (Figure 7). 

Figure 7: Top Portion of Error Report 
 

Discussing the details of the error and quarterly reports is beyond the scope of this presentation. The reports are 
generated in three formats: as a MS Office/Excel spreadsheet, as an HTML document that is displayed to the user, 
and as a PDF document that can be send to the user via e-mail. The required CPQCC error and quarterly reports had 
a very specific format that could not easily be implemented in ODS. It was easier to design a “skeleton” excel report 
certain parts of which are populated via DDE. The “skeleton” incorporates all needed special formats such that only 
data have to be added via DDE. To convert an Excel report into a PDF document, it is necessary to convert the report 
into a postscript file that is then run through the Adobe Acrobat Distiller. The postscript file is generated from within 
SAS using the %generatePS macro shown below. 
 

%MACRO generatePS(hospno,which,local,quarter); 
%GLOBAL CohortYear cpqccroot; 
DATA _NULL_; 
     FILE toexcel; 
     PUT '[open("' "&cpqccroot\DB Reports\create PS reports.xls" '")]'; 
     PUT '[WORKBOOK.SELECT("Parameters")]'; 
     PUT '[SELECT("R2C2")]'; 
     PUT '[FORMULA("' "&hospno" '")]'; 
     PUT '[SELECT("R5C2")]'; 
     PUT '[FORMULA("' "&local" '")]'; 
%IF &which EQ QTR %THEN %DO; 
     PUT '[SELECT("R3C2")]'; 

     PUT '[FORMULA("' "&CohortYear" '")]'; 
     PUT '[SELECT("R4C2")]'; 
     %IF &quarter EQ 4 %THEN %DO; PUT '[FORMULA("4")]'; %END; 
     %ELSE %IF &quarter EQ 3 %THEN %DO; PUT '[FORMULA("3")]'; %END; 
     %ELSE %IF &quarter EQ 2 %THEN %DO; PUT '[FORMULA("2")]'; %END; 
     %ELSE %IF &quarter EQ 1 %THEN %DO; PUT '[FORMULA("1")]'; %END; 
%END; 
     PUT '[run("gen' "&which" 'PS")]'; 
     PUT '[file.close(true)]'; 
RUN; 
%MEND generatePS; 
 

The macro is used to generate the PDF version of both, the error report and the quarterly report depending upon the 
parameter &which that has to be supplied when the macro is called. The line PUT '[run("gen' "&which" 
'PS")]'; calls a Visual Basic macro depending upon which report should be generated.  
 

Sub genERRPS() 
' 
' genERRPS Macro 
' 10/28/2004 by Beate Danielsen 
' 
 
' 
Range("B2").Select 
hospno = ActiveCell.FormulaR1C1 
Range("B5").Select 
localIndicator = ActiveCell.FormulaR1C1 
If localIndicator = 0 Then 
   mypath = "c:\projects\cpqccBETA\" 
Else 
mypath = "c:\projects\cpqcc\BETA\" 
End If 
ChDir mypath & "DB reports\ERR" 
Workbooks.Open Filename:= _ 
    mypath & "DB reports\ERR\CNTR_" & hospno & ".xls" 
Application.ActivePrinter = "HP DeskJet 1200C/PS on FILE:" 
ActiveWorkbook.PrintOut Copies:=1, Collate:=True, PrintToFile:=True, PrToFileName:= 
_ 
mypath & "DB reports\ERR\in\CNTR_" & hospno & ".ps" 
ActiveWorkbook.Close 
End Sub 

 
The Visual Basic macro generates a postscript file by printing to the HP DeskJet 1200C postscript printer using the 
FILE port. Adobe Acrobat Distiller is set up to convert any postscript placed in the \ERR\in folder to PDF. All these 
steps are performed on the application server when the user sends a request for an error report and do not require 
CPQCC staff. The quarterly reports are compiled four times a year, and then placed on the web server as static 
content. 

SEND DATA/REPORTS VIA E-MAIL 
As all reports are generated in real-time on the application server, and as the application server is a different physical 
location and not on the same network as the web server, it was necessary to set up a mechanism that allows hospital 
members to receive copies of the PDF versions of the error report. Besides the reports, it is also possible for each 
member hospital, to send a copy of their own most recent data as submitted to CPQCC to their hospital’s designated 
data contact person. The format of the data can be specified by the user as one of MS Office/Access 97, MS 
Office/Access 2000, comma separated ASCII file, or SAS file. Anything that is sent via e-mail is encrypted, 
compressed and password protected using the macro %sendEmailCenter. 
 

/* Send an automatic e-mail with requested item center's designated contact */ 
%MACRO sendEMailCenter(hospno); 
%GLOBAL whichDocument cpqccEmail CPQCCDataEmail dbformat byears errorcode 
        email sal name hlist0 hlist1 hlist2; 
LIBNAME contacts ODBC DATASRC="CPQCC Contacts"; 
DATA email; 
     SET contacts.hospital (KEEP=HospitalId SB_Salutation SB_Last_Name  
                                 SB_Email RC_Salutation RC_Last_Name RC_Email); 

RUN; 
LIBNAME contacts; 
%LET errorcode=0; 
%IF &whichDocument EQ QTRreport %THEN %LET contactType=RC; 
%ELSE %IF &whichDocument EQ report %THEN %LET contactType=SB; 
%ELSE %IF &whichDocument EQ master %THEN %LET contactType=SB; 
DATA _NULL_; 
      SET email (WHERE=(HospitalId EQ &hospno)); 
      CALL SYMPUT('email',TRIM(LEFT(&contactType._Email))); 
      CALL SYMPUT('sal'  ,TRIM(LEFT(&contactType._Salutation))); 
      CALL SYMPUT('name' ,TRIM(LEFT(&contactType._Last_Name))); 
 RUN; 
 %IF %BQUOTE(&email) EQ %BQUOTE() %THEN %DO; 
     %LET errorcode=1; 
     %IF &whichDocument EQ QTRreport %THEN %LET hlist1 = &hlist1 &hospno; 
     %ELSE %IF &whichDocument EQ report %THEN %LET hlist1 = &hlist1 &hospno; 
     %ELSE %IF &whichDocument EQ master %THEN %LET hlist1 = &hlist1 &hospno; 
 %END; 
 %IF &errorcode EQ 0 %THEN %DO; 
     %LET pw=%STR(); 
     DATA _NULL_; 
          SET cpqcc.mypw (KEEP=hospno EDSpw WHERE=(hospno=&hospno) READ=xx); 
          CALL SYMPUT('pw',TRIM(LEFT(pw))); 
     RUN; 
     %IF &pw EQ %STR() %THEN %DO; 
          %LET errorcode=2; 
          %IF &whichDocument EQ QTRreport %THEN %LET hlist2 = &hlist2 &hospno; 
          %ELSE %IF &whichDocument EQ report %THEN %LET hlist2 = &hlist2 &hospno; 
          %ELSE %IF &whichDocument EQ master %THEN %LET hlist2 = &hlist2 &hospno; 
     %END; 
     %ELSE %DO; 
          %IF &whichDocument EQ report %THEN %DO; 
          X "&wzpath\winrar m -p&pw -afzip -inul -ep  
             &cpqccroot\DBrepo~1\ERR\CNTR_&hospno..zip  
             &cpqccroot\DBrepo~1\ERR\out\CNTR_&hospno..pdf"; 
          FILENAME out email FROM="diane@cpqcc.org" TO=("&email")  
             SUBJECT="Center &hospno CPQCC Error Report" cc=("&cpqccemail") 
             attach="&cpqccroot\DBrepo~1\ERR\CNTR_&hospno..zip"; 
          %LET hlist0 = &hlist0 &hospno; 
          %END; 
          [... more statements to get quarterly report/data ready for e-mail ...] 
     %END; 
 %END; 
 %IF &errorcode EQ 0 %THEN %DO; 
 DATA _NULL_; 
     FILE out; 
     %IF &whichDocument EQ QTRreport %THEN %DO; 
     %IF %BQUOTE(&sal) NE %BQUOTE() AND %BQUOTE(&name) NE %BQUOTE() %THEN %DO; 
     PUT "Dear &sal &name.:"; 
     PUT " "; 
     %END; 
     %ELSE %DO; 
     put 'Dear CPQCC Member:'; 
     PUT " "; 
     %END; 
     PUT "We are pleased to present to you ..." @@; 
     [ ... more PUT statements that form the body of e-mail message ...] 
     PUT " "; 
     PUT "Sincerely, "; 
     PUT " "; 
     PUT "The CPQCC Data Center"; 
     %END; 
     [... body of e-mail message for quarterly report/data ...] 
RUN; 

QUIT; 
%END; 
%MEND sendEMailCenter; 

 

DATA CENTER STAFF SPECIFIC FEATURES 
For CPQCC Data Center staff, upon logging onto the website, additional functionalities were implemented, e.g., the 
generation of a network report summarizing the number, completeness, and errors of all current submissions, overall 
and by member center, or the generation of application server activity report. For instance, the code snippet below is 
used to determine how many requests were handled by the application server during the last 15 minutes. 
 

X "copy /Y &logfilefolder\%inputLog &logfilefolder\activeLog.log"; 
DATA _NULL_; 
     INFILE "&logfilefolder\activeLog.log" ls=93 LRECL=93 PAD; 
     INPUT @1 myweekday $ 1-3 @; 
     IF myweekday IN ('Mon' 'Tue' 'Wed' 'Thu' 'Fri' 'Sat' 'Sun') THEN DO; 
     INPUT; 
     FILE "&logfilefolder\activeLogRED.log"; 
  PUT _INFILE_; 
     END; 
RUN; 
DATA appserveruse; 
     INFILE "&logfilefolder\activeLogRED.log" TRUNCOVER SCANOVER; 
     INPUT @'ManageData' @; 
     INPUT @1 actiondate $CHAR16. @18 actiontime TIME12.; 
     starttime = INTNX('MINUTE',time(),-15); 
     IF actiontime > starttime THEN OUTPUT; 
RUN; 
DATA _NULL_; 
     IF 0 THEN SET appserveruse NOBS=nobs; 
     CALL SYMPUT('appserveruse',nobs); 
     STOP; 
RUN; 
 

WWW-BASED REPORTING 
The CPQCC Data Center’s annual paper reports were replaced by an on-line report mechanism. A demonstration of 
the reports is available at http://www.cpqccreport.org with the username cpqcc0000 and the password test. 
Three components are currently available separately for babies weighing 1,500 grams or less (small babies) and 
babies weighing over 1,500 grams (large babies).  

1. Tabulations of variables on which CPQCC collects information. An example for the form of tabulation is 
displayed in Figure 8. A hospital’s data are compared to its data for the last year. They are also compared to 
all data for all CPQCC participants. A chart in the last column of each such table allows a quick assessment 
of areas in which the hospital has percentages lower or higher than the lower and upper quartile observed 
across all CPQCC members. These tables are produced using SAS and stored as static information on the 
web server. 

2. Tabulations of standardized data allowing comparisons that control for important variations in a case mix a 
hospital treats. An example for this type of table is shown in Figure 9. If after controlling for case mix, there is 
a significant difference between a hospital’s percentage and the average network percentage, the result is 
highlighted. These tables are produced using SAS and stored as static information on the web server. 

3. Figures for all variables collected by CPQCC. An example for a dynamic chart is displayed in Figure 10. 
Each type of figure can be customized as the options in the navigation bar to the left indicate. These figures 
are generated in real-time using SAS/IntrNet. 

Figure 8: Example of Report Table 
 

 
Figure 9: Example of Report Standardized Table 

CONCLUSION 
CPQCC was able to successfully replace their previous paper-based and electronic data based system by a web-
based data base management and reporting mechanism that has been well received with CPQCC member hospitals.  
Continuing the old way would have meant hiring 2 additional staff members, purchasing an additional Teleform 
license, and using voluminous paper and mail supplies. Moreover, as the data collection instrument expanded greatly 
in 2005, both in the number of items that are collected and the number of babies eligible at each CPQCC member 
hospital, resources at CPQCC participants would have had to be increased as well.  
Converting to the web-based system channeled resources into the purchase of a SAS/IntrNet license, a web server, 
and an application server instead. CPQCC staff was successfully trained to guide the transition to a web-based 
system. On-line data collection for all centers has fully started earlier this year, and so far the results are promising in 
that fewer submissions per hospital are needed to obtain clean and usable data. Furthermore, all data appears to be 
submitted in a much timelier manner compared to previous years. Web-based reporting has turned into an important 
asset for CPQCC member centers.  
Most importantly, CPQCC can now focus on its prime initial goals, namely using the data collected in their research to 
provide the best possible NICU care in California hospitals. 

Figure 10: Dynamic Generation and Customization of Chart 

